
CORSO DI LAUREA MAGISTRALE IN INGEGNERIA MECCANICA 

PROGETTAZIONE ASSISTITA DA COMPUTER 

Costruire un modello FEM per l’analisi dello stato di tensione agente nei due CILINDRI FORZATI, mostrati in 
Figura 1, con e senza la pressione esterna. 
Il modello può essere costruito utilizzando elementi piani  (Es. PLANE42)  in  formulazione assialsimmetrica 
ed elementi GAP (Es. 169 e 171) per rappresentare il forzamento iniziale. 
Si rappresenti l’andamento delle principali componenti di tensione (assiale, radiale e circonferenziale) lungo 
il raggio, utilizzando il comando PATH. 
Dati del problema: 

• Modulo di Young = 210000 MPa 
• Raggio interno=1000 mm 
• Raggio Intermedio = 1200 mm 
• Raggio esterno = 1400 mm 
• Pressione interna = 10 MPa 
• Pressione esterna = 0 
• Interferenza radiale = 1 mm 

Nella Fig. 2 è mostrato  il modello, nella 3  la struttura deformata per  l’interferenza  iniziale, nella 4 e 5 un 
esempio di andamento delle tensioni. Infine, nel seguito è riportato un possibile file di ingresso. 

FILE DI INGRESSO  

c*************************************************** 
C*** 
C*** CILINDRI FORZATI 
C*** 
C*** USO ELEMENTI GAP, COMANDI PATH 
C*** 
FINISH 
/CLEAR 
*AFUN,DEG 
/PREP7 
C*** 
C*** PARAMETRI 
C*** 
RI0=1000  !RAGGIO INTERNO 
RIN=1200  !RAGGIO DI CONTATTO 
RES=1400  !RAGGIO ESTERNO 
 
INT=1  !INTERFERENZA RADIALE 
 
PIN=100   !PRESSIONE INTERNA 
PES=0    !PRESSIONE ESTERNA 
 
L=1000    !LUNGHEZZA ASSIALE MODELLO 
 
ESZ=25    !DIMENSIONE ELEMENTI 
 
C*** 
C*** MODELLO SOLIDO 
C*** 
RECTNG,RI0,RIN,0,L 
RECTNG,RIN,RES,0,L 
 
C*** 
C*** MATERIALE 
C*** 
MP,EX,1,210000 
 
C*** 
C*** ELEMENTI 
C*** 
et,1,42,,,1 
 
et,2,169 
et,3,171 
KEYOPT,3,9,4 
 
R,1, 
RMORE,,,,INT 
 
C*** 
C*** VINCOLI 
C*** 
LSEL,,LOC,Y,‐1,0.001 
DL,ALL,,UY,0 
ALLS 
C*** 



C*** MESH 
C*** 
type,1 
LESIZE,ALL,ESZ 
MESHKEY,1 
MSHAPE,0 
AMESH,ALL 
 
ASEL,,AREA,,1,1 
LSLA 
LSEL,R,LOC,X,RIN‐0.001,RIN+0.001 
NSLL,,1 
TYPE,2 
REAL,1 
ESURF 
 
ASEL,,AREA,,2,2 
LSLA 
LSEL,R,LOC,X,RIN‐0.001,RIN+0.001 
NSLL,,1 
TYPE,3 
REAL,1 
ESURF 
 
ALLS 
C*** 
C*** SOLUZIONE 
C*** 
/SOLU 
SOLVE 
 
LSEL,,LOC,X,RI0‐0.01,RI0+0.01 
SFL,ALL,PRESS,PIN 
 
LSEL,,LOC,X,RES‐0.01,RES+0.01 
SFL,ALL,PRESS,PES 
ALLS 
SOLVE 
 
C*** 
C*** POST‐PROCESSING 
C*** 
/POST1 
 
C*** 
C*** VISUALIZZAZIONE PRIMO CALCOLO, SOLO INTERFERENZA 
C*** 
SET,1 
/TITLE,SENZA PRESSIONE 
PATH,PIPPO,2,,300 
PPATH,1,,RI0,L/2,0 
PPATH,2,,RES,L/2,0 
PDEF,SRAD,S,X 
PDEF,SAXI,S,Y 
PDEF,SCIR,S,Z,NOAV 
PLPATH,SRAD,SCIR,SAXI 
*ASK,IFL,RETURN PER CONTINUARE,0 
 



C*** 
C*** VISUALIZZAZIONE SECONDO CALCOLO, INTERFERENZA + PRESSIONE 
C*** 
SET,2 
/TITLE,CON PRESSIONE 
PATH,PIPPO1,2,,300 
PPATH,1,,RI0,L/2,0 
PPATH,2,,RES,L/2,0 
PDEF,SRAD,S,X 
PDEF,SAXI,S,Y 
PDEF,SCIR,S,Z,NOAV 
PLPATH,SRAD,SCIR,SAXI 
 

 


